Derivadas parciales y extremos relativos

En esta práctica vas a comprobar que el programa *Mathematica* es de gran utilidad para calcular extremos relativos de funciones de varias variables. Con su ayuda podrás calcular fácilmente los puntos críticos y estudiar el carácter de la forma cuadrática asociada a la matriz hessiana en dichos puntos. Antes que nada, tenemos que aprender a calcular derivadas parciales y, de paso, aprovecharemos la ocasión para hacer con *Mathematica* algunos cálculos complicados usando la regla de la cadena para calcular derivadas parciales de segundo orden.

Derivadas

El comando `D[f[x],x,n]` calcula la derivada de orden `n` de la función `f[x]`. Naturalmente, `n` debe tener un valor concreto, pues en otro caso *Mathematica* dará un resultado en forma simbólica. No es imprescindible haber definido previamente una función para usar este comando. Puedes escribir, por ejemplo, `D[x^-9,{x,2}]` para obtener la segunda derivada de $x \mapsto x^{-9}$; y para calcular su valor cuando $x = 5$, puedes escribir `D[x^-9,{x,2}]/. x->5`. También puedes usar la notación de Lagrange `f'[x]`, `f''[x]` para obtener respectivamente las derivadas primera y segunda de `f[x]`. Lógicamente, en este caso, debes haber definido previamente la función `f`. La ventaja de esto es que con `f''[5]` obtienes el valor de `f''` para $x = 5$.

Para campos escalares de varias variables (esto es, funciones que "salen" de $\mathbb{R}^n$ y "llegan" a $\mathbb{R}$), el comando `D[f[x,y,...],{x,n1},{y,n2},...]` nos devuelve la derivada parcial de `f` respecto de `x` `n1`-veces, respecto de `y` `n2`-veces,... pudiendo omitirse el número de veces si éste es 1. Si se omite una variable quiere decir que no se deriva respecto de ella ninguna vez.

Calcula, por ejemplo, $\dfrac{\partial^3 f}{\partial x^2 \partial y}$ y $\dfrac{\partial^2 f}{\partial x \partial y}$, donde $f(x,y) = \operatorname{sen} x \cos y^2$. Usa una regla de sustitución `/.{x->Pi/4,y->`$\sqrt{\texttt{Pi/4}}$`}`, para evaluar dicha derivada en $(\pi/4, \sqrt{\pi/4})$.

```
In[1]:=
 f[x_,y_]=Sin[x] Cos[y^2]; D[f[x,y],x,y]
 D[f[x,y],{x,2},{y,2}]
```

De esta forma, para calcular el gradiente de una función de varias variables, basta construir el vector formado por las derivadas parciales primeras de la función respecto de todas las variables:

```
In[2]:=
 gradf[x_,y_]={D[f[x,y],x],D[f[x,y],y]}
```

El hessiano se calcula como sigue:

```
In[3]:=
 Hf[x_,y_]={{D[f[x,y],x,x], D[f[x,y],x,y]},
 {D[f[x,y],x,y],D[f[x,y],y,y]}}
```

Recuerda que la orden `MatrixForm[%]` permite ver la matriz anterior de la forma usual.

Plano tangente

Ya que sabemos calcular derivadas parciales de una función vamos a representar gráficamente el plano tangente a su gráfica en un punto y veremos cómo la ecuación de dicho plano se obtiene a partir de las derivadas parciales de la función. Consideremos la función

```
In[4]:=
 f[x_,y_]=1-(y^2+x)/(1+x^2)
```

y vamos a calcular en el punto (a,b) sus derivadas parciales. Por definición, la derivada parcial de f respecto a la primera variable en dicho punto viene dada por

$$\frac{\partial f}{\partial x}(a,b) = \lim_{x\to a} \frac{f(x,b)-f(a,b)}{x-a}$$

Fíjate que lo que se hace es derivar en el punto $x=a$ la función "parcial" $x \mapsto f(x,b)$, que se obtiene haciendo $y=b$ en $f(x,y)$. La gráfica de la función $z=f(x,b)$, es el conjunto $\{(x,b,f(x,b)) : x\in\mathbb{R}\}$. Observa que dicha gráfica es la curva en $\mathbb{R}^3$ que se obtiene como la intersección de la gráfica de f, $\{(x,y,f(x,y) : (x,y)\in\mathbb{R}^2\}$, con el plano de ecuación $y=b$. El vector tangente a dicha curva en el punto (a,b) es $(1,0,\frac{\partial f}{\partial x}(a,b))$. De forma análoga se interpreta la derivada parcial de f respecto a la segunda variable. Dibujemos las gráficas de la función f, de las funciones parciales $x\mapsto f(x,b)$, $y\mapsto f(a,y)$, y de sus vectores tangentes.

```
In[5]:=
 surfgraf=Plot3D[f[x,y],{x,-2,2},{y,-2,2}]

In[6]:=
 graf1[a_]:=ParametricPlot3D[{a,y,f[a,y]},{y,-2,2},
 DisplayFunction->Identity]/.
 Line[x_]->{RGBColor[0,1,0],Thickness[.007],Line[x]}

In[7]:=
 graf2[b_]:=ParametricPlot3D[{x,b,f[x,b]},{x,-2,2},
 DisplayFunction->Identity ]/.
 Line[x_]->{RGBColor[1,0,0],Thickness[.007],Line[x]}

In[8]:=
 Needs["Graphics`PlotField3D`"]

In[9]:=
 vectores[a_,b_]:=ListPlotVectorField3D[{{{a,b,
 f[a,b]},{1,0,D[f[x,y],x]/.{x->a,y->b}}},
 {{a,b,f[a, b]},{0,1,D[f[x, y],y]/.{x->a,y->b}}}},
 VectorHeads->True,
 ScaleFunction->(.8 &),ColorFunction->(RGBColor[0,0,1]&),
 DisplayFunction->Identity]/.
 Line[x_]->{Thickness[.007],Line[x]};
```

```
In[10]:=
 Show[graf1[-.8],DisplayFunction->$DisplayFunction]


In[11]:=
 Show[graf1[-.8],surfgraf,DisplayFunction->$DisplayFunction]


In[12]:=
 Show[graf2[-.5],DisplayFunction->$DisplayFunction]


In[13]:=
 Show[graf2[-.5],surfgraf,DisplayFunction->$DisplayFunction]


In[14]:=
 Show[graf1[-.8],graf2[-.5],surfgraf,
 DisplayFunction->$DisplayFunction]

In[15]:=
 Show[vectores[-.8,-.5],graf1[-.8],graf2[-.5],
 DisplayFunction->$DisplayFunction]
```

Puedes elegir otros valores para (a,b) para ver las curvas correspondientes. Naturalmente, el plano tangente a la gráfica de f en $(a,b,f(a,b))$ es el plano engendrado por los vectores tangentes a las curvas que hemos dibujado - $(1,0,\frac{\partial f}{\partial x}(a,b))$ y $(0,1,\frac{\partial f}{\partial y}(a,b))$ - y que pasa por dicho punto. Por tanto su ecuación cartesiana viene dada por:

$$\begin{vmatrix} x-a & y-b & z-f(a,b) \\ 1 & 0 & \frac{\partial f}{\partial x}(a,b) \\ 0 & 1 & \frac{\partial f}{\partial y}(a,b) \end{vmatrix} = 0$$

Es decir,

$$z-f(a,b) = \frac{\partial f}{\partial x}(a,b)(x-a) + \frac{\partial f}{\partial y}(a,b)(y-b).$$

Para acabar nos falta dibujar el plano tangente y unirlo con la gráfica anterior:

```
In[16]:=
  planotg[a_,b_]:=Plot3D[Evaluate[f[a,b]+D[f[s,t],s](x-a)+
  D[f[s,t],t](y-b)/.{s->a,t->b}],{x, a-.5,a+.5},{y,b-.5,b+.5}
  DisplayFunction->Identity]

In[17]:=
  todo[a_,b_]:=Show[surfgraf,graf1[a],graf2[b],planotg[a,b],
  vectores[a,b], DisplayFunction->$DisplayFunction]

In[18]:=
  todo[-.8,-.5]
```

Si lo piensas un poco verás que es muy fácil modificar el procedimiento anterior de forma que en vez de derivadas parciales de f en (a,b) obtengamos derivadas direccionales de f en (a,b).

"Campos" de vectores

Como sabes, el gradiente de un campo escalar f en cada punto indica la dirección de mayor aumento del campo. Como ya sabes, el comando `ContourPlot` dibuja las curvas de nivel de una función y, como vimos en clase, el gradiente es perpendicular a las mismas. Vamos a ver esto en un caso particular. Necesitaremos algunos comandos que no funcionan directamente en *Mathematica*, y para poder usarlos hay que cargar el paquete `PlotField`. Para dibujar las curvas de nivel usamos la opción `ColorFunction->Hue` que utiliza colores para diferenciar en lugar de la escala de grises usual.

```
In[19]:=
Needs["Graphics`PlotField3D`"]

In[20]:=
f[x_,y_]=x^2 + y ^2

In[21]:=
Plot3D[f[x,y],{x,-1,1},{y,-1,1}]

In[22]:=
curvnivl=ContourPlot[f[x,y],{x,-1,1},{y,-1,1},
ColorFunction->Hue]
```

La orden `PlotGradientField` dibuja los vectores $(\frac{\partial f}{\partial x}, \frac{\partial f}{\partial y})$. Las dos opciones que aparecen (`ScaleFunction` y `ScaleFactor`) no son necesarias.

```
In[23]:=
grad=PlotGradientField[f[x,y],{x,-1,1},{y,-1,1},
ScaleFunction->(.1#&),ScaleFactor->None]
```

También existe la orden `PlotHamiltonianField` que dibuja los vectores $(-\frac{\partial f}{\partial y}, \frac{\partial f}{\partial x})$. Estos son perpendiculares al gradiente y por tanto nos indican las curvas de nivel.

```
In[24]:=
ortgrad=PlotHamiltonianField[f[x,y],{x,-1,1},{y,-1,1},
ScaleFunction->(.1#&),ScaleFactor->None]

In[25]:=
Show[curvnivl,grad,ortgrad]
```

Ejercicio 1. Calcular las derivadas parciales de:

1. $f(x,y,z) = x^{y+z}, \ \forall x \in \mathbb{R}^+, y,z \in \mathbb{R}$

2. $f(x,y,z) = (x+y)^z, \ \forall x,y \in \mathbb{R}^+, z \in \mathbb{R}$

3. $f(x,y) = \operatorname{sen}(x \operatorname{sen} y), \ \forall x,y \in \mathbb{R}$

Ejercicio 2. Sea $f : \mathbb{R}^2 \setminus \{(0,0)\} \longrightarrow \mathbb{R}$ dada por $f(x,y) = \log(x^2+y^2)$ para todo $(x,y) \neq (0,0)$. Se pide:

(I) Calcúlese el gradiente de f en todo punto así como la matriz hessiana.

(II) Compruébese que

$$\frac{\partial^2 f}{\partial x^2}(x,y) + \frac{\partial^2 f}{\partial y^2}(x,y) = 0 \quad \forall (x,y) \in \mathbb{R}^2 \setminus \{(0,0)\}.$$

Ejercicio 3. Sea $f : \mathbb{R}^2 \to \mathbb{R}$ una función de clase C^2. Pongamos $g(\rho,\vartheta) = f(\rho\cos\vartheta, \rho\operatorname{sen}\vartheta)$. Comprobar que

$$\frac{\partial^2 f}{\partial x^2}(\rho\cos\vartheta,\rho\operatorname{sen}\vartheta) + \frac{\partial^2 f}{\partial y^2}(\rho\cos\vartheta,\rho\operatorname{sen}\vartheta) = \frac{\partial^2 g}{\partial \rho^2}(\rho,\vartheta) + \frac{1}{\rho^2}\frac{\partial^2 g}{\partial \vartheta^2}(\rho,\vartheta) + \frac{1}{\rho}\frac{\partial g}{\partial \rho}(\rho,\vartheta)$$

Extremos relativos

Los resultados que conocemos sobre extremos relativos de campos escalares, nos aseguran que dichos extremos son puntos críticos. Una vez calculados éstos estudiaremos la matriz hessiana en ellos, viendo si es definida, indefinida, o semidefinida. Para ello podemos usar el criterio de los valores propios: si todos son del mismo signo, la matriz es definida y hay extremo; si aparecen valores propios de distinto signo es indefinida y hay punto de silla; en otro caso, si algún valor propio es cero, la matriz es semidefinida y en tal caso no disponemos de ningún criterio que nos informe de si el punto crítico es o no es extremo relativo.

Ejemplo 1: Calculemos los extremos relativos de la función

$$f(x,y) = x^3 + 3xy^2 - 15x - 12y.$$

Primero definimos la función, su gradiente y su hessiano:

```
In[26]:=
 f[x_,y_]=x^3 + 3 x y^2 -15 x -12 y;
 gradf[x_,y_]={D[f[x,y],x],D[f[x,y],y]};
 Hf[x_,y_]={{D[f[x,y],x,x], D[f[x,y],x,y]},
 {D[f[x,y],x,y],D[f[x,y],y,y]}};
```

Para calcular los puntos críticos podemos usar el comando `Solve`, ya que el gradiente está formado por polinomios de grado bajo; asignaremos un nombre `pcrit` al resultado que usaremos para obtener los valores propios de la matriz hessiana en los puntos críticos obtenidos.:

```
In[27]:=
 pcrit=Solve[gradf[x,y]==0, {x,y}]


In[28]:=
 Eigenvalues[Hf[x,y]] /.pcrit
```

Hemos obtenido una lista formada por cuatro parejas de valores propios de `H[x,y]` correspondientes a los cuatro puntos críticos. Así, la matriz hessiana en el primer punto crítico, $(-2,-1)$, tiene sus dos valores propios negativos, con lo que es definida negativa y por tanto hay *máximo*; los puntos críticos 2.º y 3.º, $(-1,-2)$ y $(1,2)$, son *puntos de silla*, ya que en ellos la matriz hessiana es indefinida; por último, el 4.º es un punto de *mínimo*. Si queremos saber lo que vale la función f sobre sus puntos críticos, basta escribir `f[x,y]/.pcrit` lo que nos da una lista con los valores de f en los puntos críticos.

Podemos resumir diciendo que la función f tiene máximo valor relativo 28 en el punto $(-2,-1)$, mínimo valor relativo -28 en $(2,1)$, y dos puntos de silla en $(-1,-2)$ y $(1,2)$.

Ejemplo 2: Extremos relativos de

$$f(x,y) = (x^2+3y^2)\,\mathrm{e}^{1-x^2-y^2}.$$

Comenzamos definiendo f, y calculando el gradiente y la matriz hessiana:

```
In[29]:=
 Clear["Global`*"]
 f[x_,y_]=(x^2+ 3 y^2) Exp[1-x^2-y^2];
 gradf[x_,y_]={D[f[x,y],x],D[f[x,y],y]}


In[30]:=
 Hf[x_,y_]={{D[f[x,y],x,x], D[f[x,y],x,y]},
 {D[f[x,y],x,y],D[f[x,y],y,y]}}
```

Si intentamos calcular los puntos críticos resolviendo el sistema $grad f[x,y]==0$ con `NSolve`, *Mathematica* informa que no puede resolverlo, pues aparecen exponenciales, y `NSolve` (igual que `Solve` o `Reduce`) sólo sirven para polinomios y funciones sencillas. En este punto, tenemos dos posibilidades: la primera, trazar la gráfica de f (como superficie o con curvas de nivel) para tener una idea de dónde están los extremos, y usar el comando `FindRoot` con las aproximaciones iniciales que nos da la gráfica.

```
In[31]:=
 Plot3D[f[x,y],{x,-2,2},{y,-2,2}]


In[32]:=
 ContourPlot[f[x,y],{x,-2,2},{y,-2,2}]
```

A la vista de las gráficas podemos calcular aproximaciones a los extremos y calcularlos con `FindRoot`, pero ¿qué nos asegura que no hay más extremos que no entran en el rango de la función, o que no vemos por la imprecisión de los gráficos?.

La segunda posibilidad es simplificar el sistema de ecuaciones `gradf[x,y]==0`. Observamos que basta multiplicar el sistema por $e^{-(1-x^2-y^2)}$ (que no se anula), para obtener un nuevo sistema equivalente de ecuaciones polinómicas que ya sabemos cómo resolver.

```
In[33]:=
 G[x_,y_]=Simplify[gradf[x,y]*Exp[x^2+y^2-1]]

In[34]:=
 pcrit=Solve[G[x,y]==0,{x,y}]

In[35]:=
 Eigenvalues[Hf[x,y]]/.pcrit
```

Tenemos dos puntos de máximo: $(0,-1)$ y $(0,1)$; un punto de mínimo: $(0,0)$; y dos puntos de silla: $(-1,0)$ y $(1,0)$. Los valores de f en los puntos críticos los obtienes con `f[x,y]/.pcrit` con lo que, vale 3 en los dos máximos y 0 en el mínimo.

Ejercicio. Calcular los extremos relativos de las siguientes funciones.

1. $f(x,y) = x^2 + y^2 - 2x + 4y + 20$.
2. $f(x,y) = x^3 + x^2y + y^2 + 2y + 5$.
3. $f(x,y) = (x^2 + y^2)\,e^{x^2-y^2}$.
4. $f(x,y,z) = x^2 + y^2 + z^2 - 2x$.
5. $f(x,y) = x^3y^3 - y^4 - x^4 + xy$.